Pensieve header: A unified verification notebook for the sl_2-portfolio project; continues pensieve://Projects/PPSA/nb/Verification.pdf.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Portfolio"];
NotebookOpen[wdir <> "\\MakeSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization / Utilities

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$p = 2; $k = 2; (* $k can't be ∞ at least because of Faddeev-Quesne. *)
If[$k == 0, ϵ = 0, ϵ /: ϵ^k_ /; k > $k := 0]; (* $k=0 fails in Series[..{ϵ,...}] *)
SetAttributes[{SS, SST}, HoldAll];
TRule = {T_i_ → ℯ^(ℏ t_i), T → ℯ^(ℏ t)};
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
   Collect[Normal@Series[ℰ, {ℏ, 0, $p}], ℏ, Together] ];
SST[ℰ_] :=
  Block[{ℏ, ϵ}, Collect[Normal@Series[ℰ /. TRule, {ℏ, 0, $p}], ℏ, Together] ];
Simp[ℰ_, op_] := Collect[ℰ, _CU | _QU, op];
Simp[ℰ_] := Simp[ℰ, Collect[Normal@Series[#, {ℏ, 0, $p}], ℏ, Expand] &];
SimpT[ℰ_] :=
  Collect[ℰ, _CU | _QU, Collect[Normal@Series[# /. TRule, {ℏ, 0, $p}], ℏ, Expand] &];
```

Differential polynomials (DP):

Utils

```
DP_{α_→D_x_, β_→D_y_}[P_][λ_] :=
 Total[CoefficientRules[P, {α, β}] /. ({m_, n_} → c_) :→ c D[λ, {x, m}, {y, n}]]
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, CE, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts} },
   (#_U = U@#) & /@ gs;
   gp = Alternatives @@ gs; gp = gp | gp_; (* gens *)
   sr = Thread[gs → Range@Length@gs]; (* sorting → *)
   cp = Alternatives @@ cs; (* cents *)
   CE[ℰ_] := Collect[ℰ, _U, (Expand[#] /. ħ^d_ /; d > $p :→ 0) &];
   U_i_[ℰ_] := ℰ /. {t : cp :→ t_i, u_U :→ Replace[u, x_ :→ x_i, 1]};
   U_i_[NCM[]] = pow[ℰ_, 0] = U@{} = 1_U = U[];
   B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
   B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
   B[U@y_, U@x_] := CE[-B[U@x, U@y]];
   x_ ** 1_U := x; 1_U ** x_ := x;
   (a_. * x_U) ** (b_. * y_U) := If[a b === 0, 0, CE[a b (x ** y)]];
   U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
     U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
   U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
   U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
   U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
   U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
   U@{l_, r___} := U@{Expand[l], r};
   U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
   𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
     sp = Replace[{specs}, l_List :→ l_null, {1}];
     vs = Join @@ (First /@ sp);
     us = Join @@ (sp /. l_s_ :→ (l /. x_i_ :→ x_s));
     CE[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) :→ c U@(us^p)
       ]] /. x_null :→ x
    ];
   pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
   𝕊_U[ℰ_, ss___Rule] := CE@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) :→ c NCM @@ MapThread[pow, {Last /@ {ss}, p}]];
   S_i_[c_. * u_U] := CE[(c /. S_i[U, Centrals]) DeleteCases[u, _i] **
      U_i[NCM @@ Reverse@Cases[u, x_i :→ S@U@x]]]; ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) :→ (m[U[g]] = img), {1}];
  m[1_U] = 1_V;
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U :→ m[u]]; )
```

Meta-Operations

QLImplementation

```
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[a_CU, y_CU] = -γ y_CU; B[x_CU, a_CU] = -γ x_CU;
B[x_CU, y_CU] = 2 ϵ a_CU - t 1_CU;
(S@CU@y = -y_CU; S@a_CU = -a_CU; S@x_CU = -x_CU;)
S_i_[CU, Centrals] = {t_i → -t_i};
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas}]]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{2.0625,
 {(28 t^2 γ^4 + 116 t γ^5 ϵ) CU[y, y, y, x, x] + <<21>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Verifying that *S* is an anti-homomorphism on CU:

```
With[{bas = CU /@ {y_1, a_1, x_1}},
 Table[HL@Simp[S_1[z1 ** z2] - S_1[z2] ** S_1[z1]],
  {z1, bas}, {z2, bas}]]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying the involutivity of *S* on products of triples:

```
With[{bas = CU /@ {y_1, a_1, x_1}},
 Table[HL@Simp[z1 ** z2 ** z3 - S_1@S_1[z1 ** z2 ** z3]],
  {z1, bas}, {z2, bas}, {z3, bas}]]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Implementing QU

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[ⅇ^(γ ϵ ℏ)];
B[a_QU, y_QU] = -γ y_QU; B[x_QU, a_QU] = -γ QU@x;
B[x_QU, y_QU] = (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T ⅇ^(-2 ϵ a ℏ)) / ℏ], {a}];
(S@y_QU = 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) y], {a, y}]; S@a_QU = -a_QU; S@x_QU = 𝕆_QU[SS[-ⅇ^(ℏ ϵ a) x], {a, x}];)
S_i_[QU, Centrals] = {t_i → -t_i, T_i → T_i^-1};
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$\{\{\{QU[y], QU[y]\} \to 0,\ \{QU[y], QU[a]\} \to \gamma\, QU[y],$
$\{QU[y], QU[x]\} \to \frac{(-1+T)\, QU[\,]}{\hbar} - 2\, T\, \epsilon\, QU[a] - \gamma\, \epsilon\, \hbar\, QU[y, x]\},$
$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y],\ \{QU[a], QU[a]\} \to 0,\ \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$
$\{\{QU[x], QU[y]\} \to \frac{(1-T)\, QU[\,]}{\hbar} + 2\, T\, \epsilon\, QU[a] + \gamma\, \epsilon\, \hbar\, QU[y, x],$
$\{QU[x], QU[a]\} \to -\gamma\, QU[x],\ \{QU[x], QU[x]\} \to 0\}\}$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}

Verifying associativity on a “random” triple (~34 secs @ $p=5, $k=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp]
  }] // Timing
```

$\{18.6875,\ \{\left(\frac{28\,\gamma^4 - 56\,T\,\gamma^4 + 28\,T^2\,\gamma^4}{\hbar^2} + \frac{82\,\gamma^5\,\epsilon - 280 \ll 2 \gg \epsilon + 198\,T^2\,\gamma^5\,\epsilon}{\hbar}\right) QU[y, y, y, x, x] +$
$\ll 18 \gg + (1 + 8\,\gamma\,\epsilon\,\hbar)\, QU[\ll 1 \gg],\ 0\}\}$

Verifying that *S* is an anti-homomorphism on QU:

```
With[{bas = QU /@ {y1, a1, x1}},
 Table[{z1, z2} → HL@Simp[S1[z1 ** z2] - S1[z2] ** S1[z1]],
  {z1, bas}, {z2, bas} ] ]
```

$\{\{\{QU[y_1], QU[y_1]\} \to 0,\ \{QU[y_1], QU[a_1]\} \to 0,\ \{QU[y_1], QU[x_1]\} \to 0\},$
$\{\{QU[a_1], QU[y_1]\} \to 0,\ \{QU[a_1], QU[a_1]\} \to 0,\ \{QU[a_1], QU[x_1]\} \to 0\},$
$\{\{QU[x_1], QU[y_1]\} \to 0,\ \{QU[x_1], QU[a_1]\} \to 0,\ \{QU[x_1], QU[x_1]\} \to 0\}\}$

Verifying that $\mathrm{Lim}_{\hbar \to 0}$ QU = CU using a “random” product (~23 secs @ $p=5, $k=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. TRule ⋃ {QU → CU}, ℏ → 0] - lhs] // HL
  }] // Timing
```

$\{28.1875, \{48\, t\, \gamma^5 \in \mathrm{CU}[y, y, y, x, x] + \ll 77 \gg + \mathrm{CU}[y, y, y, y, y, a, a, a, a, x, x, x, x],$
$2 \left(\frac{4 \gamma^5 \in}{\hbar} - \frac{8 T \gamma^5 \in}{\hbar} + \frac{4 T^2 \gamma^5 \in}{\hbar}\right) \mathrm{QU}[y, y, y, x, x] +$
$\ll 217 \gg + 8 \gamma \in \hbar\, \mathrm{QU}[y, y, y, y, y, a, a, a, a, x, x, x, x], 0\}\}$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -x_CU, a → -a_CU, x → -y_CU}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1/2 ℯ^(ℏ ϵ a) x], {a, x}],
  a → -a_QU, x → 𝕆_QU[SS[-T^-1/2 ℯ^(ℏ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Simp[Qθ[Qθ[z]], PowerExpand]], {z, bas}] ]
```

$\left\{\mathrm{QU}[y] \to -\frac{\mathrm{QU}[x]}{\sqrt{T}} - \frac{\in \hbar\, \mathrm{QU}[a, x]}{\sqrt{T}} \to \mathrm{QU}[y],\ \mathrm{QU}[a] \to -\mathrm{QU}[a] \to \mathrm{QU}[a],\right.$
$\left.\mathrm{QU}[x] \to \left(-\frac{1}{\sqrt{T}} + \frac{\gamma \in \hbar}{\sqrt{T}}\right) \mathrm{QU}[y] - \frac{\in \hbar\, \mathrm{QU}[y, a]}{\sqrt{T}} \to \mathrm{QU}[x]\right\}$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

```
A𝔻$f = γ ((Cosh[ℏ (a e + γ e/2 - t/2)] - Cosh[ℏ Sqrt[((t - γ e)/2)^2 + e ω]]) /
    (ℏ ⅇ^(ℏ ((a+γ) e-t/2)) Sinh[γ e ℏ/2] (a^2 e + a γ e - a t - ω)));
```

Scaling behaviour of A𝔻$f:

```
HL@Simplify[A𝔻$f == ((A𝔻$f /. γ → 1) /. {e → γ e, a → γ^-1 a, ω → γ^-1 ω})]
True
```

```
HL@FullSimplify[
  A𝔻$f == ((A𝔻$f /. γ → 1) /. {ℏ → γ^2 ℏ, e → e / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
True
```

ADeq

```
A𝔻$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[A𝔻, QU → CU,
  {a → a_CU, x → CU@x, y → 𝕊_CU[SS[A𝔻$f] /. e → ϵ, a → a_CU, ω → A𝔻$ω] ** y_CU}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[SimpT[A𝔻[z1 ** z2] - A𝔻[z1] ** A𝔻[z2]]], {z1, bas}, {z2, bas}] ]
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

```
S𝔻$g = Sqrt[((2 γ (Cosh[ℏ/2 Sqrt[t^2 + γ^2 e^2 + 4 e ϖ]] - Cosh[(t - e γ - 2 e a)/(2/ℏ)]))/
    (Sinh[γ e ℏ/2] (t (2 a + γ) - 2 a (a + γ) e + 2 ϖ) ℏ))];
```

Verify agreement with the formulas in pensieve://People/VanDerVeen/Dequant1.pdf:

```
{SD$P = (Cosh[ħ ((e-t)/2 + e a)] - Cosh[ħ Sqrt[(t^2+e^2)/4 + e ϖ]]) / (ħ Sinh[(-e ħ)/2] (ϖ - e a^2 + (t - e) a + t/2)),
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ħ → γ^2 ħ, e → e/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
    SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 FullSimplify[SD$g == FullSimplify[
    Sqrt[SD$Q] /. c → a + 1/2 /. {ħ → γ^2 ħ, e → e/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$\{-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,e+\frac{e-t}{2}\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}\left(e^2+t^2\right)+e\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{e\,\hbar}{2}\right]\right)\Big/\left(\left(-a^2 e+\frac{t}{2}+a\,(-e+t)+\varpi\right)\hbar\right)\right)$, True, True,

$\left(4\left(-\mathrm{Cosh}\left[\frac{1}{2}\sqrt{e^2+t^2+4\,e\,\varpi}\;\hbar\right]+\mathrm{Cosh}\left[c\,e\,\hbar-\frac{t\,\hbar}{2}\right]\right)\mathrm{Csch}\left[\frac{e\,\hbar}{2}\right]\right)\Big/\left(\left(\left(-1+4\,c^2\right)e-4\,(c\,t+\varpi)\right)\hbar\right)$,
True, True}

SDeq

```
SD$f = Simplify[ⅇ^(ħ (t/2-e a)) (SD$g /. {a → -a, t → -t})];
```

SDeq

```
SD$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ 1_CU / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → a_CU,
  x → S_CU[SS[SD$f] /. e → ϵ, a → a_CU, ϖ → SD$ϖ] ** x_CU,
  y → S_CU[SS[SD$g] /. e → ϵ, a → a_CU, ϖ → SD$ϖ] ** y_CU }]
```

Verifying the θ-symmetry:

```
Table[HL@SimpT[Cθ[SD[z]] == SD[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

{True, True, True}

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@SimpT[SD[z1 ** z2] - SD[z1] ** SD[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

R in QU.

Faddeev-Quesne's formula:

Quesne

$$\mathbb{e}_{q\_,k\_}[x\_] := \mathbb{e}\hat{}\left(\sum_{j=1}^{k} \frac{(1-q)^j x^j}{j\,(1-q^j)}\right);\ \mathbb{e}_{q\_}[x\_] := \mathbb{e}_{q,\$k}[x]$$

```
Table[Together@SeriesCoefficient[e_{ρ,5}[x], {x, 0, n}], {n, 0, 5}]
```

$$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)}, 1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[e_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
```

{1, 1, 1, 1, 1, 1}

R

$$QU[R_{i\_,j\_}] := \mathbb{O}_{QU}\left[SS\left[\mathbb{e}^{\hbar b_1 a_2}\,\mathbb{e}_q[\hbar y_1 x_2] \,/.\, b_1 \to \gamma^{-1}(\epsilon a_1 - t_i)\right], \{y_1, a_1\}_i, \{a_2, x_2\}_j\right];$$
$$QU\left[R^{-1}_{i\_,j\_}\right] := S_j@QU[R_{i,j}];$$

```
QU[R_{3,4}] // Short
```

$$QU[\,] + \frac{\epsilon\,\hbar\,QU[a_3, a_4]}{\gamma} + \hbar\,QU[y_3, x_4] + \frac{\ll 1 \gg}{\gamma} + \ll 1 \gg - \ll 1 \gg - \frac{\epsilon \ll 2 \gg \ll 1 \gg}{\gamma^2} - \frac{\hbar^2\,QU[y_3, a_4, x_4]\,t_3}{\gamma} + \frac{\hbar^2\,QU[a_4, a_4]\,t_3^2}{2\,\gamma^2}$$

Verifying R2 (~2 secs @ $p=4, $k=2):

```
QU[R_{1,2} ** R^{-1}_{1,2}] // Simp // HL // Timing
```

{0.140625, QU[]}

Verifying R3 (~156 secs @ $p=4, $k=2):

```
{Short[lhs = QU[R_{1,2} ** R_{1,3} ** R_{2,3}]], HL@SimpT[lhs - QU[R_{2,3} ** R_{1,3} ** R_{1,2}]]} // Timing
```

$$\left\{1.01563, \left\{QU[\,] + \frac{\epsilon\,\hbar\,QU[a_1, a_2]}{\gamma} + \ll 85 \gg + QU[y_1, y_1, x_3, x_3]\left(\frac{\hbar^2}{2} - \hbar^2 T_2 + \frac{1}{2}\hbar^2 T_2^2\right), 0\right\}\right\}$$

The representation ρ

rho

```
ρ@y_CU = ρ@y_QU = (0 0; ϵ 0); ρ@a_CU = ρ@a_QU = (γ 0; 0 0);
ρ@x_CU = (0 γ; 0 0); ρ@x_QU = SS@(0 (1 - e^(-γ ϵ ℏ)) / (ϵ ℏ); 0 0);
ρ[e^ℰ_] := MatrixExp[ρ[ℰ]];
ρ[ℰ_] :=
  (ℰ /. {t → γ ϵ, T → e^(ℏ γ ϵ)} /. (U : CU | QU) [u___] :→ Fold[Dot, (1 0; 0 1), ρ /@ U /@ {u}])
```

Verifying that ρ represents CU and QU:

```
Table[ρ[z1 ** z2] == ρ[z1].ρ[z2] // SS // HL,
 {U, {CU, QU}}, {z1, U /@ {y, a, x}}, {z2, U /@ {y, a, x}} ]
```

```
{{{True, True, True}, {True, True, True}, {True, True, True}},
 {{True, True, True}, {True, True, True}, {True, True, True}}}
```

The Logoi Λ

Goal. In either U, compute $F = e^{-\eta y}\, e^{\xi x}\, e^{\eta y}\, e^{-\xi x}$. First compute $G = e^{\xi x}\, y\, e^{-\xi x}$, a finite sum. Now F satisfies the ODE $\partial_\eta F = \partial_\eta\left(e^{-\eta y}\, e^{\eta G}\right) = -yF + FG$ with initial conditions $F(\eta = 0) = 1$. So we set it up and solve:

```
With[{U = CU},
 Module[{G, F, fs, bs, e, b, es, sol},
  G = Echo@Simp[Table[ξ^k / k!, {k, 0, $k + 1}].NestList[Simp[B[x_U , #]] &, y_U, $k + 1]];
  fs = Echo@Flatten@Table[f_{l,i,j,k}[η], {l, 0, $k}, {i, 0, 1}, {j, 0, 1}, {k, 0, 1}];
  F = Echo[fs.(bs = fs /. f_{l_,i_,j_,k_}[η] :→ ϵ^l U@{y^i, a^j, x^k})];
  es = Flatten[
    Table[Coefficient[e, b] == 0, {e, {F - 1_U /. η → 0, F ** G - y_U ** F - ∂_η F}}, {b, bs}]];
  sol = Echo@First[F /. DSolve[es, fs, η]];
  Echo[sol /. {e- → 1, U → Times}];
  Collect[sol /. {e- → 1, U → Times}, ϵ, Simplify]
 ]]
```

$$-t\,\xi\,CU[\,] + 2\,\epsilon\,\xi\,CU[a] - \gamma\,\epsilon\,\xi^2\,CU[x] + CU[y]$$

$$\{f_{0,0,0,0}[\eta],\ f_{1,0,0,0}[\eta],\ f_{1,0,0,1}[\eta],\ f_{1,0,1,0}[\eta],\ f_{1,0,1,1}[\eta],\ f_{1,1,0,0}[\eta],\ f_{1,1,0,1}[\eta],\ f_{1,1,1,0}[\eta],\ f_{1,1,1,1}[\eta],\ f_{2,0,0,0}[\eta],\ f_{2,0,0,1}[\eta],\ f_{2,0,0,2}[\eta],\ f_{2,0,1,0}[\eta],\ f_{2,0,1,1}[\eta],\ f_{2,0,1,2}[\eta],\ f_{2,0,2,0}[\eta],\ f_{2,0,2,1}[\eta],\ f_{2,0,2,2}[\eta],\ f_{2,1,0,0}[\eta],\ f_{2,1,0,1}[\eta],\ f_{2,1,0,2}[\eta],\ f_{2,1,1,0}[\eta],\ f_{2,1,1,1}[\eta],\ f_{2,1,1,2}[\eta],\ f_{2,1,2,0}[\eta],\ f_{2,1,2,1}[\eta],\ f_{2,1,2,2}[\eta],\ f_{2,2,0,0}[\eta],\ f_{2,2,0,1}[\eta],\ f_{2,2,0,2}[\eta],\ f_{2,2,1,0}[\eta],\ f_{2,2,1,1}[\eta],\ f_{2,2,1,2}[\eta],\ f_{2,2,2,0}[\eta],\ f_{2,2,2,1}[\eta],\ f_{2,2,2,2}[\eta]\}$$

$$\begin{aligned}&CU[\,]\,f_{0,0,0,0}[\eta] + \epsilon\,CU[\,]\,f_{1,0,0,0}[\eta] + \epsilon\,CU[x]\,f_{1,0,0,1}[\eta] + \epsilon\,CU[a]\,f_{1,0,1,0}[\eta] + \epsilon\,CU[a, x]\,f_{1,0,1,1}[\eta] +\\
&\epsilon\,CU[y]\,f_{1,1,0,0}[\eta] + \epsilon\,CU[y, x]\,f_{1,1,0,1}[\eta] + \epsilon\,CU[y, a]\,f_{1,1,1,0}[\eta] + \epsilon\,CU[y, a, x]\,f_{1,1,1,1}[\eta] +\\
&\epsilon^2\,CU[\,]\,f_{2,0,0,0}[\eta] + \epsilon^2\,CU[x]\,f_{2,0,0,1}[\eta] + \epsilon^2\,CU[x, x]\,f_{2,0,0,2}[\eta] + \epsilon^2\,CU[a]\,f_{2,0,1,0}[\eta] +\\
&\epsilon^2\,CU[a, x]\,f_{2,0,1,1}[\eta] + \epsilon^2\,CU[a, x, x]\,f_{2,0,1,2}[\eta] + \epsilon^2\,CU[a, a]\,f_{2,0,2,0}[\eta] + \epsilon^2\,CU[a, a, x]\,f_{2,0,2,1}[\eta] +\\
&\epsilon^2\,CU[a, a, x, x]\,f_{2,0,2,2}[\eta] + \epsilon^2\,CU[y]\,f_{2,1,0,0}[\eta] + \epsilon^2\,CU[y, x]\,f_{2,1,0,1}[\eta] + \epsilon^2\,CU[y, x, x]\,f_{2,1,0,2}[\eta] +\\
&\epsilon^2\,CU[y, a]\,f_{2,1,1,0}[\eta] + \epsilon^2\,CU[y, a, x]\,f_{2,1,1,1}[\eta] + \epsilon^2\,CU[y, a, x, x]\,f_{2,1,1,2}[\eta] +\\
&\epsilon^2\,CU[y, a, a]\,f_{2,1,2,0}[\eta] + \epsilon^2\,CU[y, a, a, x]\,f_{2,1,2,1}[\eta] + \epsilon^2\,CU[y, a, a, x, x]\,f_{2,1,2,2}[\eta] +\\
&\epsilon^2\,CU[y, y]\,f_{2,2,0,0}[\eta] + \epsilon^2\,CU[y, y, x]\,f_{2,2,0,1}[\eta] + \epsilon^2\,CU[y, y, x, x]\,f_{2,2,0,2}[\eta] +\\
&\epsilon^2\,CU[y, y, a]\,f_{2,2,1,0}[\eta] + \epsilon^2\,CU[y, y, a, x]\,f_{2,2,1,1}[\eta] + \epsilon^2\,CU[y, y, a, x, x]\,f_{2,2,1,2}[\eta] +\\
&\epsilon^2\,CU[y, y, a, a]\,f_{2,2,2,0}[\eta] + \epsilon^2\,CU[y, y, a, a, x]\,f_{2,2,2,1}[\eta] + \epsilon^2\,CU[y, y, a, a, x, x]\,f_{2,2,2,2}[\eta]\end{aligned}$$

$$\begin{aligned}&e^{-t\eta\xi}\,CU[\,] + \frac{1}{2}e^{-t\eta\xi}\,t\,\gamma\,\epsilon\,\eta^2\,\xi^2\,CU[\,] + \frac{1}{24}e^{-t\eta\xi}\,t\,\gamma^2\,\epsilon^2\,\eta^3\,\xi^3\,(-8+3\,t\,\eta\,\xi)\,CU[\,] +\\
&2\,e^{-t\eta\xi}\,\epsilon\,\eta\,\xi\,CU[a] + e^{-t\eta\xi}\,\gamma\,\epsilon^2\,\eta^2\,\xi^2\,(-1+t\,\eta\,\xi)\,CU[a] - e^{-t\eta\xi}\,\gamma\,\epsilon\,\eta\,\xi^2\,CU[x] -\\
&\frac{1}{2}e^{-t\eta\xi}\,\gamma^2\,\epsilon^2\,\eta^2\,\xi^3\,(-2+t\,\eta\,\xi)\,CU[x] - e^{-t\eta\xi}\,\gamma\,\epsilon\,\eta^2\,\xi\,CU[y] - \frac{1}{2}e^{-t\eta\xi}\,\gamma^2\,\epsilon^2\,\eta^3\,\xi^2\,(-2+t\,\eta\,\xi)\,CU[y] +\\
&2\,e^{-t\eta\xi}\,\epsilon^2\,\eta^2\,\xi^2\,CU[a, a] - 2\,e^{-t\eta\xi}\,\gamma\,\epsilon^2\,\eta^2\,\xi^3\,CU[a, x] + \frac{1}{2}e^{-t\eta\xi}\,\gamma^2\,\epsilon^2\,\eta^2\,\xi^4\,CU[x, x] -\\
&2\,e^{-t\eta\xi}\,\gamma\,\epsilon^2\,\eta^3\,\xi^2\,CU[y, a] + e^{-t\eta\xi}\,\gamma^2\,\epsilon^2\,\eta^3\,\xi^3\,CU[y, x] + \frac{1}{2}e^{-t\eta\xi}\,\gamma^2\,\epsilon^2\,\eta^4\,\xi^2\,CU[y, y]\end{aligned}$$

$$\begin{aligned}&1 + 2\,a\,\epsilon\,\eta\,\xi - y\,\gamma\,\epsilon\,\eta^2\,\xi - x\,\gamma\,\epsilon\,\eta\,\xi^2 + \frac{1}{2}t\,\gamma\,\epsilon\,\eta^2\,\xi^2 + 2\,a^2\,\epsilon^2\,\eta^2\,\xi^2 - 2\,a\,y\,\gamma\,\epsilon^2\,\eta^3\,\xi^2 +\\
&\frac{1}{2}y^2\,\gamma^2\,\epsilon^2\,\eta^4\,\xi^2 - 2\,a\,x\,\gamma\,\epsilon^2\,\eta^2\,\xi^3 + x\,y\,\gamma^2\,\epsilon^2\,\eta^3\,\xi^3 + \frac{1}{2}x^2\,\gamma^2\,\epsilon^2\,\eta^2\,\xi^4 - \frac{1}{2}y\,\gamma^2\,\epsilon^2\,\eta^3\,\xi^2\,(-2+t\,\eta\,\xi) -\\
&\frac{1}{2}x\,\gamma^2\,\epsilon^2\,\eta^2\,\xi^3\,(-2+t\,\eta\,\xi) + a\,\gamma\,\epsilon^2\,\eta^2\,\xi^2\,(-1+t\,\eta\,\xi) + \frac{1}{24}t\,\gamma^2\,\epsilon^2\,\eta^3\,\xi^3\,(-8+3\,t\,\eta\,\xi)\end{aligned}$$

$$\begin{aligned}&1 + \frac{1}{2}\epsilon\,\eta\,\xi\,\left(4\,a + \gamma\,\left(-2\,y\,\eta - 2\,x\,\xi + t\,\eta\,\xi\right)\right) + \frac{1}{24}\epsilon^2\,\eta^2\,\xi^2\,\Big(48\,a^2 - 24\,a\,\gamma\,\left(1 + 2\,y\,\eta + 2\,x\,\xi - t\,\eta\,\xi\right) +\\
&\quad\gamma^2\,\Big(12\,y^2\,\eta^2 - 12\,y\,\eta\,\left(-2 - 2\,x\,\xi + t\,\eta\,\xi\right) + \xi\,\left(12\,x^2\,\xi - 12\,x\,\left(-2 + t\,\eta\,\xi\right) + t\,\eta\,\left(-8 + 3\,t\,\eta\,\xi\right)\right)\Big)\Big)\end{aligned}$$

Logos

```
λ[U_] := Module[{G, F, fs, f, bs, e, b, es},
   G = Simp[Table[ξ^k / k!, {k, 0, $k + 1}].NestList[Simp[B[x_U, #]] &, y_U, $k + 1]];
   fs = Flatten@Table[f_{l,i,j,k}[η], {l, 0, $k}, {i, 0, l}, {j, 0, l}, {k, 0, l}];
   F = fs.(bs = fs /. f_{l_,i_,j_,k_}[η] :> ϵ^l U@{y^i, a^j, x^k});
   es = Flatten[
      Table[Coefficient[e, b] == 0, {e, {F - 1_U /. η → 0, F ** G - y_U ** F - ∂_η F}}, {b, bs}]];
   Collect[First[F /. DSolve[es, fs, η] /. {e- → 1, U → Times}], ϵ, Simplify]];
```

```
λ[CU]
```

$$1 + \frac{1}{2}\epsilon\,\eta\,\xi\,\left(4\,a + \gamma\,\left(-2\,y\,\eta - 2\,x\,\xi + t\,\eta\,\xi\right)\right)$$

```
λ[QU]
```

$$1+\frac{1}{4\hbar}\epsilon\,\eta\,\xi\,\left(8\,a\,T\,\hbar+(-1+3\,T)\,\gamma\,\eta\,((-1+T)\,\xi-2\,y\,\hbar)+2\,x\,\gamma\,\hbar\,(\xi-3\,T\,\xi+2\,y\,\hbar)\right)$$

Logos

```
wc[CU] = t; wc[QU] = (T - 1) / ℏ;
Λ[U_] := Λ[U] = Module[{Q, w}, Q = (-w ξ η + η y + ξ x + δ y x) / (1 + w δ);
    Collect[(1 + w δ)^-1 ⅇ^-Q DP_{ξ→D_x, η→D_y}[λ[U]][ⅇ^Q] /. w → wc[U], ϵ, Simplify]];
Λ[U_, t1_, T1_, y1_, a1_, x1_, ξ1_, η1_, δ1_] :=
  Λ[U] /. {t → t1, T → T1, y → y1, a → a1, x → x1, ξ → ξ1, η → η1, δ → δ1};
```

```
Λ[CU]
```

$$\frac{1}{1+t\,\delta}+\frac{1}{2\,(1+t\,\delta)^5}\epsilon\,\Big(4\,a\,(1+t\,\delta)^2\,((t+x\,y)\,\delta^2+\eta\,\xi+\delta\,(1+y\,\eta+x\,\xi))+\gamma\,\big(2\,t^3\,\delta^4+4\,t^2\,\delta^2\,(\delta-x\,y\,\delta^2+\eta\,\xi)-2\,(y\,\eta\,(\delta\,(2+y\,\eta)+\eta\,\xi)+x^2\,\delta\,(2\,y^2\,\delta^2+3\,y\,\delta\,\xi+\xi^2)+x\,(3\,y^2\,\delta^2\,\eta+4\,y\,\delta\,(\delta+\eta\,\xi)+\xi\,(2\,\delta+\eta\,\xi)))-t\,(3\,x^2\,y^2\,\delta^4-4\,\delta\,\eta\,\xi-\eta^2\,\xi^2+4\,x\,y\,\delta^3\,(3+y\,\eta+x\,\xi)+\delta^2\,(-2+y^2\,\eta^2+4\,x\,\xi+x^2\,\xi^2+4\,y\,(\eta+x\,\eta\,\xi)))\big)\Big)$$

```
Λ[QU]
```

$$\frac{\hbar}{(-1+T)\,\delta+\hbar}+\frac{1}{4\,((-1+T)\,\delta+\hbar)^5}\epsilon\,\hbar^2\,\Big(8\,a\,T\,((-1+T)\,\delta+\hbar)^2\,(\eta\,\xi\,\hbar+\delta\,(1+y\,\eta+x\,\xi)\,\hbar+\delta^2\,(-1+T+x\,y\,\hbar))+\gamma\,\big(\eta\,\xi\,\hbar^2\,((-1+3\,T)\,\eta\,((-1+T)\,\xi-2\,y\,\hbar)+2\,x\,\hbar\,(\xi-3\,T\,\xi+2\,y\,\hbar))+(-1+T)\,\delta^4\,(-2+6\,T^3-x^2\,y^2\,\hbar^2-2\,T^2\,(7+4\,x\,y\,\hbar)+T\,(10+8\,x\,y\,\hbar-5\,x^2\,y^2\,\hbar^2))-4\,\delta^3\,\hbar\,(1-3\,T^3+x^2\,y^2\,\hbar^2+T^2\,(7+2\,x\,y\,(3+y\,\eta)\,\hbar+2\,x^2\,y\,\xi\,\hbar)+T\,(-5-2\,x\,y\,(3+y\,\eta)\,\hbar+x^2\,y\,\hbar\,(-2\,\xi+y\,\hbar)))+2\,\delta\,\hbar^2\,((1-3\,T)\,y^2\,\eta^2\,\hbar+2\,\eta\,(\xi+3\,T^2\,\xi-4\,T\,\xi\,(1+x\,y\,\hbar)+y\,\hbar\,(1-3\,T+x\,y\,\hbar))+x\,\hbar\,((x-3\,T\,x)\,\xi^2+2\,y\,\hbar+\xi\,(2-6\,T+2\,x\,y\,\hbar)))-\delta^2\,\hbar\,((1-4\,T+3\,T^2)\,y^2\,\eta^2\,\hbar+\hbar\,(-2+3\,T^2\,(-2+4\,x\,\xi+x^2\,\xi^2)+4\,x\,(\xi+y\,\hbar)+x^2\,(\xi^2+2\,y\,\xi\,\hbar-4\,y^2\,\hbar^2)-2\,T\,(-4+x\,(8\,\xi-6\,y\,\hbar)+x^2\,\xi\,(2\,\xi-5\,y\,\hbar)))+2\,\eta\,(-2\,(-1+T)\,\xi\,(1+3\,T^2-2\,T\,(2+x\,y\,\hbar))+y\,\hbar\,(2+6\,T^2+x\,y\,\hbar+T\,(-8+5\,x\,y\,\hbar))))\big)\Big)$$

```
{Short[lhs = O_CU[SS[ⅇ^(ℏ (ξ x+η y+δ x y))], {x, y}], 5], HL@Simp[lhs -
    O_CU[SS[ⅇ^(ℏ (ξ x+η y+δ x y-t ℏ ξ η)/(1+ℏ t δ)) Λ[CU, t, T, y, a, x, ℏ ξ, ℏ η, ℏ δ]], {y, a, x}], Together]}
```

$$\Big\{(1-t\,\delta\,\hbar+t^2\,\delta^2\,\hbar^2+t\,\gamma\,\delta^2\,\epsilon\,\hbar^2-t\,\eta\,\xi\,\hbar^2-t^3\,\delta^3\,\hbar^3-3\,t^2\,\gamma\,\delta^3\,\epsilon\,\hbar^3+2\,t^2\,\delta\,\eta\,\xi\,\hbar^3+2\,t\,\gamma\,\delta\,\epsilon\,\eta\,\xi\,\hbar^3)\,\mathrm{CU}[\,]+(2\,\delta\,\epsilon\,\hbar-4\,t\,\delta^2\,\epsilon\,\hbar^2+2\,\epsilon\,\eta\,\xi\,\hbar^2+6\,t^2\,\delta^3\,\epsilon\,\hbar^3-8\,t\,\delta\,\epsilon\,\eta\,\xi\,\hbar^3)\,\mathrm{CU}[a]+(\xi\,\hbar-2\,t\,\delta\,\xi\,\hbar^2-2\,\gamma\,\delta\,\epsilon\,\xi\,\hbar^2+3\,t^2\,\delta^2\,\xi\,\hbar^3+9\,t\,\gamma\,\delta^2\,\epsilon\,\xi\,\hbar^3-t\,\eta\,\xi^2\,\hbar^3-\gamma\,\epsilon\,\eta\,\xi^2\,\hbar^3)\,\mathrm{CU}[x]+(\eta\,\hbar-2\,t\,\delta\,\eta\,\hbar^2-2\,\gamma\,\delta\,\epsilon\,\eta\,\hbar^2+3\,t^2\,\delta^2\,\eta\,\hbar^3+9\,t\,\gamma\,\delta^2\,\epsilon\,\eta\,\hbar^3-t\,\eta^2\,\xi\,\hbar^3-\gamma\,\epsilon\,\eta^2\,\xi\,\hbar^3)\,\mathrm{CU}[y]+(4\,\delta\,\epsilon\,\xi\,\hbar^2-12\,t\,\delta^2\,\epsilon\,\xi\,\hbar^3+2\,\epsilon\,\eta\,\xi^2\,\hbar^3)\,\mathrm{CU}[a,x]+\left(\frac{\xi^2\,\hbar^2}{2}-\frac{3}{2}\,t\,\delta\,\xi^2\,\hbar^3-3\,\gamma\,\delta\,\epsilon\,\xi^2\,\hbar^3\right)\mathrm{CU}[x,x]+\ll 14\gg+\frac{1}{2}\,\delta\,\eta^2\,\hbar^3\,\mathrm{CU}[y,y,y,x]+3\,\delta^3\,\epsilon\,\hbar^3\,\mathrm{CU}[y,y,a,x,x]+\frac{1}{2}\,\delta^2\,\xi\,\hbar^3\,\mathrm{CU}[y,y,x,x,x]+\frac{1}{2}\,\delta^2\,\eta\,\hbar^3\,\mathrm{CU}[y,y,y,x,x]+\frac{1}{6}\,\delta^3\,\hbar^3\,\mathrm{CU}[y,y,y,x,x,x],\ 0\Big\}$$

```
{Short[lhs = SimpT@𝕆QU[SS[ℯ^(ℏ (ξ x+η y+δ x y))], {x, y}], 5],
 rhs = SimpT@
   𝕆QU[SS[ℯ^(ℏ ν (ξ x+η y+δ x y-(T-1) ξ η)) Λ[QU, t, T, y, a, x, ℏ ξ, ℏ η, ℏ δ] /. ν → (1 + (T - 1) δ)^-1],
    {y, a, x}];
 HL[Simplify[lhs == rhs]]}
```

$\left\{\left(1-t\,\delta\,\hbar+\left(-\frac{t^2\,\delta}{2}+t^2\,\delta^2+t\,\gamma\,\delta^2\,\epsilon-t\,\eta\,\xi\right)\hbar^2\right)QU[\,]+\left(2\,\delta\,\epsilon\,\hbar+\left(2\,t\,\delta\,\epsilon-4\,t\,\delta^2\,\epsilon+2\,\epsilon\,\eta\,\xi\right)\hbar^2\right)QU[a]+\left(\xi\,\hbar+\left(-2\,t\,\delta\,\xi-2\,\gamma\,\delta\,\epsilon\,\xi\right)\hbar^2\right)QU[x]+\left(\eta\,\hbar+\left(-2\,t\,\delta\,\eta-2\,\gamma\,\delta\,\epsilon\,\eta\right)\hbar^2\right)QU[y]+4\,\delta\,\epsilon\,\xi\,\hbar^2\,QU[a,x]+\frac{1}{2}\,\xi^2\,\hbar^2\,QU[x,x]+4\,\delta\,\epsilon\,\eta\,\hbar^2\,QU[y,a]+\left(\delta\,\hbar+\left(-2\,t\,\delta^2+\gamma\,\delta\,\epsilon-4\,\gamma\,\delta^2\,\epsilon+\eta\,\xi\right)\hbar^2\right)QU[y,x]+\frac{1}{2}\,\eta^2\,\hbar^2\,QU[y,y]+4\,\delta^2\,\epsilon\,\hbar^2\,QU[y,a,x]+\delta\,\xi\,\hbar^2\,QU[y,x,x]+\delta\,\eta\,\hbar^2\,QU[y,y,x]+\frac{1}{2}\,\delta^2\,\hbar^2\,QU[y,y,x,x],\ \text{True}\right\}$

C𝕆, Q𝕆, and Swaps

Swaps from Pensieve://Talks/Toulouse-1705/DogmaDemo.nb and from Pensieve://Talks/Sydney-1708/ExtraDetails@@.nb.

CdsO

```
SetAttributes[{C𝕆, Q𝕆}, Orderless];
CU@C𝕆[specs___, 𝔼[L_, Q_, P_]] := 𝕆CU[SS[ℯ^(L+Q) P], specs];
QU@Q𝕆[specs___, 𝔼[L_, Q_, P_]] := 𝕆QU[SS[ℯ^(L+Q) P], specs];
```

```
CU@C𝕆[𝔼[ℏ t1 a2, ℏ t1^-1 (ℯ^t1 - 1) y1 x2, 1 + ϵ x1 y2], {y1, x1}1, {x2, a2, y2}2] // Short
```

$CU[\,]+\ll 13\gg+CU[y_1,x_1]\left(-\gamma\,\epsilon\,\hbar^2\,t_2+e^{t_1}\,\gamma\,\epsilon\,\hbar^2\,t_2+\frac{\epsilon\,\hbar\,t_2}{t_1}-\frac{e^{t_1}\,\epsilon\,\hbar\,t_2}{t_1}\right)$

```
HL[ρ[ℯ^(ξ CU@x)].ρ[ℯ^(α CU@a)] == ρ[ℯ^(α CU@a)].ρ[ℯ^(ℯ^(-γ α) ξ CU@x)]]
```

True

SW

```
SW_{x_i_, a_j_}[(𝒪 : C𝕆 | Q𝕆)[OrderlessPatternSequence[{lh___, x_i_, a_j_, rh___}_s_,
    more___, 𝔼[L_, Q_, P_]]]] := 𝒪[{lh, a_j, x_i, rh}_s, more,
  With[{q = ℯ^(-γ α) ξ x_i + α a_j},
   𝔼[L, ℯ^(-γ α) ξ x_i + (Q /. x_i → 0), ℯ^-q DP_{x_i→D_ξ, a_j→D_α}[P][ℯ^q]] /. {α → ∂_{a_j} L, ξ → ∂_{x_i} Q}]]
```

```
c𝕆 = C𝕆[𝔼[ℏ t1 a2, ℏ t1^-1 (ℯ^t1 - 1) y1 x2, 1 + ϵ x1 y2], {y1, x1}1, {x2, a2, y2}2]
```

$C\mathbb{O}\left[\{y_1,x_1\}_1,\{x_2,a_2,y_2\}_2,\mathbb{E}\left[\hbar\,a_2\,t_1,\frac{\left(-1+e^{t_1}\right)\hbar\,x_2\,y_1}{t_1},1+\epsilon\,x_1\,y_2\right]\right]$

```
SW_{x2,a2}[c𝕆]
```

$C\mathbb{O}\left[\{y_1,x_1\}_1,\{a_2,x_2,y_2\}_2,\mathbb{E}\left[\hbar\,a_2\,t_1,\frac{e^{-\gamma\,\hbar\,t_1}\left(-1+e^{t_1}\right)\hbar\,x_2\,y_1}{t_1},1+\epsilon\,x_1\,y_2\right]\right]$

```
With[{c𝕆 = C𝕆[{y1, x1}1, {x2, a2, y2}2, 𝔼[ℏ t1 a2, ℏ t1^-1 (ⅇ^t1 - 1) y1 x2, 1 + ϵ x1 y2]]},
 HL[CU[c𝕆] == CU[c𝕆 // SW_{x2,a2}]]]
```

True

```
With[{c𝕆 = C𝕆[{y1, a1, x1}1, {x2, a2, y2}2,
    𝔼[ℏ (l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2), ℏ (γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
     1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]]},
 {CU[c𝕆] // Short, HL[CU[c𝕆] == CU[c𝕆 // SW_{x2,a2}]]}
]
```

$\{CU[a_1, a_1, a_1] \left(\frac{1}{2} \epsilon \hbar^2 l_1 l_{11}^2 t_1^2 + \epsilon \hbar^2 l_1 l_{11} l_{21} t_1 t_2 + \frac{1}{2} \epsilon \hbar^2 l_1 l_{21}^2 t_2^2\right) + \ll 75 \gg + CU[\,] (\ll 1 \gg),$

True}

```
With[{q𝕆 = Q𝕆[{y1, a1, x1}1, {x2, a2, y2}2,
    𝔼[ℏ (l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2), ℏ (γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
     1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]]},
 {QU[q𝕆] // Short, HL[QU[q𝕆] == QU[q𝕆 // SW_{x2,a2}]]}
]
```

$\{QU[a_1, a_1, a_1] \left(\frac{1}{2} \epsilon \hbar^2 l_1 l_{11}^2 t_1^2 + \epsilon \hbar^2 l_1 l_{11} l_{21} t_1 t_2 + \frac{1}{2} \epsilon \hbar^2 l_1 l_{21}^2 t_2^2\right) + \ll 75 \gg + QU[\,] (\ll 1 \gg),$

True}

SW

```
SW_{a_j_,y_i_}[(𝒪 : C𝕆 | Q𝕆)[OrderlessPatternSequence[{lh___, a_j_, y_i_, rh___}_s_,
    more___, 𝔼[L_, Q_, P_]]]] := 𝒪[{lh, y_i, a_j, rh}_s, more,
  With[{q = ⅇ^(-γ α) η y_i + α a_j},
   𝔼[L, ⅇ^(-γ α) η y_i + (Q /. y_i → 0), ⅇ^-q DP_{y_i→D_η, a_j→D_α}[P][ⅇ^q]] /. {α → ∂_{a_j} L, η → ∂_{y_i} Q}]]
```

```
With[{q𝕆 = Q𝕆[{y1, a1, x1}1, {x2, a2, y2}2,
    𝔼[ℏ (l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2), ℏ (γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
     1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]]},
 {QU[q𝕆] // Short, HL[QU[q𝕆] == QU[q𝕆 // SW_{a2,y2}]]}
]
```

$\{QU[a_1, a_1, a_1] \left(\frac{1}{2} \epsilon \hbar^2 l_1 l_{11}^2 t_1^2 + \epsilon \hbar^2 l_1 l_{11} l_{21} t_1 t_2 + \frac{1}{2} \epsilon \hbar^2 l_1 l_{21}^2 t_2^2\right) + \ll 75 \gg + QU[\,] (\ll 1 \gg),$

True}

SW

```
SW_{x_i_,y_j_→k_}[C𝕆[{lh___, x_i_, y_j_, rh___}_s_, more___, 𝔼[L_, Q_, P_]]] :=
 C𝕆[{lh, y_k, a_k, x_k, rh}_s, more,
  With[{q = ν (ξ x_k + η y_k + δ x_k y_k - t_k ξ η)},
   𝔼[L, q + (Q /. x_i | y_j → 0), ⅇ^-q DP_{x_i→D_ξ, y_j→D_η}[P][Λ[CU, t_k, T_k, y_k, a_k, x_k, ξ, η, δ] ⅇ^q]] /.
    ν → (1 + t_k δ)^-1 /. {ξ → (∂_{x_i} Q /. y_j → 0), η → (∂_{y_j} Q /. x_i → 0), δ → ∂_{x_i,y_j} Q}]]
```

```
With[{cⅅ = Cⅅ[{x1, y1}1, {x2, a2, y2}2,
     𝔼[ℏ (l12 t1 a2 + l22 t2 a2), ℏ (γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
      1 + ϵ (l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]]},
 {CU[cⅅ] // Short, HL[CU[cⅅ] == CU[cⅅ // SW_{x1,y1→1}]]}
]
```

$\{\mathrm{CU}[a_2, a_2, a_2, a_2]\left(\frac{1}{6}\epsilon\hbar^3 l_2 l_{12}^3 t_1^3 + \frac{1}{2}\epsilon\hbar^3 l_2 l_{12}^2 l_{22} t_1^2 t_2 + \frac{1}{2}\epsilon\hbar^3 l_2 l_{12} l_{22}^2 t_1 t_2^2 + \frac{1}{6}\epsilon\hbar^3 l_2 l_{22}^3 t_2^3\right) + \ll 123 \gg + \mathrm{CU}[\,]\left(\ll 1 \gg\right), \mathrm{True}\}$

SW

```
SW_{x_i_,y_j_→k_}[Qⅅ[{lh___, x_i_, y_j_, rh___}_s_, more___, 𝔼[L_, Q_, P_]]] :=
 Qⅅ[{lh, y_k, a_k, x_k, rh}_s, more,
  With[{q = ν (ξ x_k + η y_k + δ x_k y_k - ℏ^-1 (T_k - 1) ξ η)},
   𝔼[L, q + (Q /. x_i | y_j → 0), e^-q DP_{x_i→D_ξ, y_j→D_η}[P][Λ[QU, t_k, T_k, y_k, a_k, x_k, ξ, η, δ] e^q]] /.
    ν → (1 + ℏ^-1 (T_k - 1) δ)^-1 /. {ξ → (∂_{x_i} Q /. y_j → 0), η → (∂_{y_j} Q /. x_i → 0), δ → ∂_{x_i,y_j} Q}]]
```

```
With[{qⅅ = Qⅅ[{x1, y1}1, {x2, a2, y2}2,
     𝔼[ℏ (l12 t1 a2 + l22 t2 a2), ℏ (γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
      1 + ϵ (l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]]},
 {QU[qⅅ] // Short, HL[err = SimpT[QU[qⅅ] - QU[qⅅ // SW_{x1,y1→1}]]]}
]
```

$\{\frac{1}{6}\epsilon\hbar^3 \mathrm{QU}[y_1, y_1, y_1, y_1, x_1, x_1, x_1, x_1]\, p_{11}\gamma_{11}^3 + \ll 159 \gg + \mathrm{QU}[\,]$
$\left(1 + \frac{\epsilon p_{11}}{\hbar} + \frac{\epsilon p_{22}}{\hbar} - \gamma\epsilon l_{12} p_{22} t_1 + \ll 1713 \gg + \frac{\epsilon p_{11} T_1^2 T_2^6 \gamma_{22}^3}{\hbar} + 36\gamma\epsilon^2 p_{22} T_2^8 \gamma_{22}^3 + \frac{4\epsilon p_{22} T_2^8 \gamma_{22}^3}{\hbar}\right), 0\}$

Rewrite Rules

RR: Rewrite Rule. RQ: Revised Quadratic.

RR

```
RR[{u_i_, w_j_} → {vs__, k_}, {𝓋_, ω_}, RQ_, λ_][(𝒪 : Cⅅ | Qⅅ)[
    OrderlessPatternSequence[{lh___, u_i_, w_j_, rh___}_s_, more___, 𝔼[Q_, P_]]]] :=
  𝒪[{lh, Sequence @@ (#_k & /@ {vs}), rh}_s, more, 𝔼[
    (RQ /. (v : u | w | t | T) :> v_k) + (Q /. u_i | w_j → 0),
    e^-RQ DP_{u_i→D_𝓋, w_j→D_ω}[P][Λ[𝒪, t_k, T_k, y_k, a_k, x_k, 𝓋, ω, δ] e^RQ] /.
     {𝓋 → (∂_{𝓋_i} Q /. ω_j → 0), ω → (∂_{ω_j} Q /. 𝓋_i → 0), δ → ∂_{𝓋_i,ω_j} Q}
   ]];
```

𝔼

𝔼[L, Q, P] means $e^{\hbar(L+Q)}$ P, where L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial. It should be interpreted via Cⅅ[𝔼[...], $\{x_1,\ a_1,\ y_1\}_i$, ...] (with some default for direct interpretation), or likewise via Qⅅ[𝔼[...], $\{x_1,\ a_1,\ y_1\}_i$, ...]. In themselves, Cⅅ and Qⅅ should have an interpretation in CU/QU by casting.

Alternative Algorithms

Logos

```
λalt[CU] := Module[{eqn, d, b, c, sol},
   eqn = ρ[ⅇ^(ξ xCU)].ρ[ⅇ^(η yCU)] == ρ[ⅇ^(d yCU)].ρ[ⅇ^(c (t 1CU - 2 ϵ aCU))].ρ[ⅇ^(b xCU)];
   {sol} = Solve[Thread[Flatten /@ eqn], {d, b, c}] /. C[1] → 0;
   Collect[ⅇ^(-η y-ξ x+η ξ t) Normal@Series[ⅇ^(c t + d y - 2 ϵ c a + b x) /. sol, {ϵ, 0, $k}], ϵ, Simplify]];
```

```
{λalt[CU], HL[λalt[CU] == λ[CU]]}
```

$\left\{1 + \frac{1}{2} \epsilon\, \eta\, \xi\, \left(4\, a + \gamma\, \left(-2\, y\, \eta - 2\, x\, \xi + t\, \eta\, \xi\right)\right),\ \text{True}\right\}$